

matrix svd — Singular value decomposition

| | | | |
|--------------------------------|-------------------|-------------------------|----------------------|
| Syntax Methods and formulas | Menu Reference | Description Also see | Remarks and examples |
|--------------------------------|-------------------|-------------------------|----------------------|

Syntax

```
matrix svd U w V = A
```

where **U**, **w**, and **V** are matrix names (the matrices may exist or not) and **A** is the name of an existing $m \times n$ matrix, $m \geq n$.

Menu

Data > Matrices, ado language > Singular value decomposition

Description

`matrix svd` produces the singular value decomposition (SVD) of **A**.

Also see [\[M-5\] svd\(\)](#) for alternative routines for obtaining the singular value decomposition.

Remarks and examples

stata.com

The singular value decomposition of $m \times n$ matrix **A**, $m \geq n$, is defined as

$$\mathbf{A} = \mathbf{U} \text{diag}(\mathbf{w}) \mathbf{V}'$$

U: $m \times n$, **w**: $1 \times n$, $\text{diag}(\mathbf{w})$: $n \times n$, and **V**: $n \times n$, where **U** is column orthogonal ($\mathbf{U}'\mathbf{U} = \mathbf{I}$ if $m = n$), all the elements of **w** are positive or zero, and $\mathbf{V}'\mathbf{V} = \mathbf{I}$.

Singular value decomposition can be used to obtain a g2-inverse of **A** (\mathbf{A}^* : $n \times m$, such that $\mathbf{A}\mathbf{A}^*\mathbf{A} = \mathbf{A}$ and $\mathbf{A}^*\mathbf{A}\mathbf{A}^* = \mathbf{A}^*$ —the first two Moore–Penrose conditions) via $\mathbf{A}^* = \mathbf{V}\{\text{diag}(1/w_j)\}\mathbf{U}'$, where $1/w_j$ refers to individually taking the reciprocal of the elements of **w** and substituting 0 if $w_j = 0$ or is small. If **A** is square and of full rank, $\mathbf{A}^* = \mathbf{A}^{-1}$.

► Example 1

Singular value decomposition is used to obtain accurate inverses of nearly singular matrices and to obtain g2-inverses of matrices that are singular, to construct orthonormal bases, and to develop approximation matrices. Our example will prove that `matrix svd` works:

```
. matrix A = (1,2,9\2,7,5\2,4,18)
. matrix svd U w V = A
. matrix list U
U[3,3]
           c1           c2           c3
r1  .42313293  .89442719  -.1447706
r2  .3237169  -6.016e-17  .94615399
r3  .84626585  -.4472136  -.2895412
. matrix list w
w[1,3]
           c1           c2           c3
r1  21.832726  2.612e-16  5.5975071
```

```
. matrix list V
V[3,3]
      c1      c2      c3
c1   .12655765  .96974658  .2087456
c2   .29759672 - .23786237  .92458514
c3   .94626601 - .05489132 - .31869671

. matrix newA = U*diag(w)*V'
. matrix list newA
newA[3,3]
      c1  c2  c3
r1    1  2  9
r2    2  7  5
r3    2  4 18
```

As claimed, **newA** is equal to our original **A**.

The g_2 -inverse of **A** is computed below. The second element of **w** is small, so we decide to set the corresponding element of $\text{diag}(1/w_j)$ to zero. We then show that the resulting **Ainv** matrix has the properties of a g_2 -inverse for **A**.

```
. matrix Winv = J(3,3,0)
. matrix Winv[1,1] = 1/w[1,1]
. matrix Winv[3,3] = 1/w[1,3]
. matrix Ainv = V*Winv*U'
. matrix list Ainv
Ainv[3,3]
      r1      r2      r3
c1   -.0029461  .03716103  -.0058922
c2   -.0181453  .16069635  -.03629059
c3   .02658185  -.0398393  .05316371

. matrix AAiA = A*Ainv*A
. matrix list AAiA
AAiA[3,3]
      c1  c2  c3
r1    1  2  9
r2    2  7  5
r3    2  4 18

. matrix AiAAi = Ainv*A*Ainv
. matrix list AiAAi
AiAAi[3,3]
      r1      r2      r3
c1   -.0029461  .03716103  -.0058922
c2   -.0181453  .16069635  -.03629059
c3   .02658185  -.0398393  .05316371
```



Methods and formulas

Stewart (1993) surveys the contributions of five mathematicians—Beltrami, Jordan, Sylvester, Schmidt, and Weyl—who established the existence of the singular value decomposition and developed its theory.

Reference

Stewart, G. W. 1993. On the early history of the singular value decomposition. *SIAM Review* 35: 551–566.

Also see

- [P] **matrix** — Introduction to matrix commands
- [P] **matrix define** — Matrix definition, operators, and functions
- [M-4] **matrix** — Matrix functions
- [M-5] **svd()** — Singular value decomposition
- [U] **14 Matrix expressions**